数理科学 IIA / 情報数学概論 IIA (2013): 多項式剰余列の計算例

照井 章（筑波大学 数理物質系）

2013 年 7 月 31 日

例題

Donald E. Knuth, “The Art of Computer Programming”, Vol. 2: Seminumerical algorithms, p. 427 の例:

```
u = x^8 + x^6 - 3 x^4 - 3 x^3 + 8 x^2 + 2 x - 5
```

$-5+2x+8x^2-3x^3-3x^4+x^6+x^8$

```
v = 3 x^6 + 5 x^4 - 4 x^2 - 9 x + 21
```

$21-9x-4x^2+5x^4+3x^6$

関数定義

lc[p]: p の主係数 (leading coefficient) を求める

CoefficientList を用いると、多項式のすべての係数を要素にもつリストを求めることができる。（変数の次数に関して昇べきの順に並んでいることに注意。）

```
CoefficientList[u, x]
```

{-5, 2, 8, -3, -3, 0, 1, 0, 1}

主係数は CoefficientList で出力されるリストの最後の要素なので、最後の要素を取り出す関数を定義する。（–1 番目の要素＝最後の要素）

```
lc[p_] := CoefficientList[p, x][[-1]]
```

deg[p]: p の次数 (degree) を求める

Exponent で多項式の特定の変数に関する次数を得られる。

```
deg[p_] := Exponent[p, x]
```

擬除算による多項式剰余列 (PRS) の計算

$P_1 = u$, $P_2 = v$ とおく。

```
p1 = u
```

$-5 + 2x + 8x^2 - 3x^3 - 3x^4 + x^6 + x^8$

```
p2 = v
```

$21 - 9x - 4x^2 + 5x^4 + 3x^6$

ギリシャ文字の入力は、メニューの「パレット」→「その他」→「基礎的なタイプセット」によりパレットを表示させせることで入力可能であるが、ローマ字の a, b で代用させてかまわない。

```
α3 = lc[p2] ^ (deg[p1] - deg[p2] + 1)
```

27

多項式剰余の計算には PolynomialRemainder を用いる。以下、 P_3, P_4, ... と多項式剰余列の要素を計算する。

```
p3 = PolynomialRemainder[α3 * p1, p2, x]
```

$-9 + 3x^2 - 15x^4$

```
α4 = lc[p3] ^ (deg[p2] - deg[p3] + 1)
```

-3375

```
p4 = PolynomialRemainder[α4 * p2, p3, x]
```

$-59\,535 + 30\,375x + 15\,795x^2$

```
α5 = lc[p4] ^ (deg[p3] - deg[p4] + 1)
```

3 940 568 584 875

```
p5 = PolynomialRemainder[α5 * p3, p4, x]
```

$-1\,654\,608\,338\,437\,500 + 1\,254\,542\,875\,143\,750x$

```
α6 = lc[p5] ^ (deg[p4] - deg[p5] + 1)
```

1 573 877 825 573 946 701 583 164 062 500

```
p6 = PolynomialRemainder[α6 * p4, p5, x]
```

12 593 338 795 500 743 100 931 141 992 187 500

```
N[Log10[Abs[p6]]]
```

34.1001

P_6 の係数は35桁まで膨張している。

縮小PRS算法による多項式剰余列 (PRS) の計算

$P_1 = u,\ P_2 = v$ とおく。

```
p1 = u
```

$-5 + 2x + 8x^2 - 3x^3 - 3x^4 + x^6 + x^8$

```
p2 = v
```

$21 - 9x - 4x^2 + 5x^4 + 3x^6$

P_3 は通常の擬剰余として計算する。

```
α3 = lc[p2] ^ (deg[p1] - deg[p2] + 1)
```

27

```
p3 = PolynomialRemainder[α3 * p1, p2, x]
```

$-9 + 3x^2 - 15x^4$

```
α4 = lc[p3] ^ (deg[p2] - deg[p3] + 1)
```

-3375

ここで初めて縮小 PRS 算法による β_4 を計算する。

```
β4 = α3
```

27

P_4 は擬剰余を β_4 で割る。

```
p4 = PolynomialRemainder[α4 * p2, p3, x] / β4
```

$$\frac{1}{27}\left(-59\,535 + 30\,375\,x + 15\,795\,x^2\right)$$

このままでは係数が整数ではないので、Expand で展開する。

```
p4 = Expand[p4]
```

$-2205 + 1125x + 585x^2$

```
α5 = lc[p4] ^ (deg[p3] - deg[p4] + 1)
```

200 201 625

```
β5 = α4
```

-3375

```
p5 = PolynomialRemainder[α5 * p3, p4, x] / β5
```

$$\frac{84\,062\,812\,500 - 63\,737\,381\,250\,x}{3375}$$

```
p5 = Expand[p5]
```

$24\,907\,500 - 18\,885\,150\,x$

```
α6 = lc[p5]^(deg[p4] - deg[p5] + 1)
```

356 648 890 522 500

```
β6 = α5
```

200 201 625

```
p6 = PolynomialRemainder[α6 * p4, p5, x] / β6
```

527 933 700

```
N[Log10[Abs[p6]]]
```

8.72258

擬剰余による P_6 の桁数が35桁に対して、縮小 PRS では P_6 の桁数が9桁となり、1/4 程度に減少している。

ノートブックの操作

- *Mathematica* の計算を行うウインドウは、ノートブックと呼ばれる。
- ノートブック上のすべての入出力は、セルと呼ばれる単位に分けられる。
- セルには 「スタイル」 という属性がある。
- ノートブックの各セルでは、内容に応じた適切なスタイルを設定する。
  メニュー：「書式」 → 「スタイル」 → 各スタイル
- セルの選択 → セルのブラケット （右端のカッコ） をクリック。複数のセルを一度に選択可能。
- セルの削除 → セルを選択後、Deleteキーを押す。
- 式 （セル） の実行（評価）は、実行するセルにカーソルを置いてShift + Return。